# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xeyes
pkgver=1.1.1
pkgrel=3
pkgdesc="Follow the mouse/SHAPE extension X demo"
arch=('i686' 'x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libx11' 'libxt' 'libxext' 'libxmu' 'libxrender')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xeyes-${pkgver}.tar.bz2)
sha1sums=('efe6116d31a7f69e4fb6038613e52b0960b9b61c')

build() {
  cd "${srcdir}/xeyes-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/xeyes-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
